set nocount on

declare @version varchar(10)
SELECT  @version=left(convert(varchar,SERVERPROPERTY('productversion')),1)

declare @SPID bigint,
@SQL varchar(1000)

CREATE TABLE #sp_who3
( 
SPID INT, 
Status VARCHAR(200) NULL, 
Login sysname NULL, 
HostName sysname NULL, 
BlkBy sysname NULL, 
DBName sysname NULL, 
Command VARCHAR(8000) NULL, 
CPUTime bigINT NULL, 
DiskIO bigINT NULL, 
LastBatch VARCHAR(2000) NULL, 
ProgramName VARCHAR(200) NULL, 
SPID2 INT
) 
if @version>=9 or @version=1
begin
alter table #sp_who3 add requestid int null
end

CREATE TABLE #sp_who3_agg
( 
SPID INT, 
HostName sysname NULL, 
Login sysname NULL, 
ProgramName VARCHAR(200) NULL, 
CPUTime bigINT NULL, 
DiskIO bigINT NULL, 
Command VARCHAR(8000) NULL, 
buffer text null,
 
buffer2 text null
)
 

select * into #sp_who4 from #sp_who3
select * into #sp_who4_agg from #sp_who3_agg

INSERT into #sp_who3  EXEC sp_who2 'active' 
WAITFOR DELAY '00:00:10'
insert #sp_who4 exec sp_who2 'active'

update who4
set who4.login=x.login
from #sp_who4 who4
inner join #sp_who4 x
on who4.spid=x.spid
and x.login<>''

insert into #sp_who3_agg
select spid,hostname,login, programname,sum(cputime) as cputime,sum(diskio) as diskio, command, null, null
from #sp_who3
group by spid,hostname,login, programname, command

insert into #sp_who4_agg
select spid,hostname,login, programname,sum(cputime) as cputime,sum(diskio) as diskio, command, null, null
from #sp_who4
group by spid,hostname,login, programname, command

CREATE TABLE #Inputbuffer(
EventType NVARCHAR(30) NULL,
Parameters INT NULL,
EventInfo NVARCHAR(4000) NULL
)

declare xcn cursor local fast_forward read_only for
select distinct
[spid]
from
#sp_who4_agg
--where spid>50

open xcn

fetch next from xcn into @SPID

while @@fetch_status=0
begin
set @SQL='DBCC INPUTBUFFER(' + convert(varchar(25),@SPID) + ') with no_infomsgs'
insert into #Inputbuffer
exec (@SQL)
update #sp_who4_agg set #sp_who4_agg.[buffer]=convert(text,n.[EventInfo])
from
#Inputbuffer n
where
#sp_who4_agg.[spid]=@spid
truncate table #inputbuffer

fetch next from xcn into @SPID
end
close xcn
deallocate xcn

drop table #Inputbuffer

if @version>=9
begin --workaround to make code not fail on sql2000
declare @sql2 nvarchar(1000)
set @sql2=N'update who4agg
set who4agg.buffer2=dest.text
from  #sp_who4_agg who4agg inner join 
sys.dm_exec_requests as der on who4agg.spid=der.session_id
cross apply sys.dm_exec_sql_text (der.sql_handle) as dest'
exec sp_executesql @sql2
end

SELECT who3.spid,who3.hostname,who3.programname, who4.login,who4.cputime,sum(who4.cputime)-sum(who3.cputime) as cpu_diff,who4.diskio,sum(who4.diskio)-sum(who3.diskio) as diskio_diff,
who4.command,sj.name as JobName, convert(varchar(8000),who4.buffer) as buffer,convert(varchar(8000),who4.buffer2) as buffer2
FROM #sp_who3_agg who3
full outer join #sp_who4_agg who4 on who3.spid=who4.spid
left outer join msdb.dbo.sysjobs sj on right(replace(convert(varchar(36),sj.job_id),'-',''),16)=SUBSTRING(who4.programname, 48, 16)
    WHERE who3.spid > 50 
group by who3.spid,who3.hostname,who3.programname,who4.login,who4.cputime,who4.diskio,who4.command,sj.name,convert(varchar(8000),who4.buffer),convert(varchar(8000),who4.buffer2)
 
drop table #sp_who3_agg
drop table #sp_who4_agg
DROP TABLE #sp_who3
drop table #sp_who4

